
The Application of Reference-Scaled Average Bioequivalence in Bioequivalence Analysis with Highly Variable Drug

Abstract

The Average Bioequivalence (ABE) is usually used to evaluate the bioequivalence of chemical drug preparations, and the equivalent standard is that the 90% CI of the geometric mean ratio of the main pharmacokinetic parameters (AUC and C_{max}) between the test preparation and the reference preparation falls within the range of 80.00%~125.00%.

For some drugs, due to low bioavailability, acid instability, extensive metabolism before absorption and the other reasons, the within-subject coefficient of variation (CV_w) of one or more pharmacokinetic parameters is greater than or equal to 30%, called highly variable drug (HVD). Other factors being equal, the number of subjects required for bioequivalence study will increase with the increase of the within-subject coefficient of variation. For highly variable drugs, the conventional sample size and equivalence criteria are adopted, and sometimes the bioequivalence cannot be proved even when the reference drug is compared with itself.

For highly variable drugs with good safety and wide treatment window, based on full scientific demonstration and on the premise of ensuring the safety and effectiveness of drug use for the public, crossing design by partial or complete repetition, according to the within-subject coefficient of variation of the reference preparation, the reference-Scaled Average Bioequivalence (RSABE) method was proposed. At the same time, the equivalence criteria should be appropriately relaxed based on 80.00%~125.00%. In this way, the unnecessary exposure of the population can be reduced and the purpose of scientific evaluation of whether different preparations are bioequivalent can be achieved.

Apart from the consideration of clinical significance, statistical methods are particularly important in the study of bioequivalence. This paper introduces a SAS macro which can select statistical methods and evaluate the equivalence according to the within-subject coefficient of variation. In addition, the established macros are easy to extend, making it possible to handle more complex requirements.

Key words: reference-Scaled Average Bioequivalence, highly variable drug, cross designed trial, SAS macro

Introduction

Bioequivalence study usually selected 18 to 24 cases of healthy subjects, with cross experiment design, using the method of bioavailability studies, and pharmacokinetic parameters as indicators, comparing the same kind of drugs of the same or different dosage form preparation, under the same experimental conditions, to judge the statistically significant difference of active ingredient absorption degree and speed. Statistical analysis after the subjects took the test preparation and the reference preparation, the pharmacokinetic parameters of the two drugs, such as AUC and C_{max} , were log-transformed and analyzed by ANOVA for significance test. Then, the two one-sided t-test and 90% confidence interval were used to evaluate and judge the bioequivalence between the two preparations with the range of 80.00%~125.00%.

For the highly variable drug (HVD), in the evaluation of bioequivalence by using the usual 18-24 cases crossover design and equivalence judgment criteria, the degree of assurance is

reduced due to the increase of intra-individual differences, which is likely to lead to class II errors in statistics. As a result, the test preparation that is actually bioequivalent to the reference drug is judged to be not equivalent.

Various methods are proposed to minimize the bias of bioequivalence evaluation by NMPA. This paper mainly introduces the repeated crossover design and the application of RSABE method with the innovative SAS macro. The SAS macro statistical methods supports a cross design was designed for three cycles of partial repetition (only reference preparation was reused) or four cycles of complete repetition (repeated use of reference preparation and test preparation), and flexibly preprocess various BDS data.

Procedure

Step 1: Determine S_{wr} , the within-subject standard deviation of the reference product, for the pharmacokinetic parameters AUC and C_{max} .

Step 2: Find least squares geometric mean, ratio of least squares geometric mean, and 90% CI of GMR when $S_{wr} < 0.294$.

Step 3: Get 95% upper confidence bound, and find least squares geometric mean, and ratio of least squares geometric mean when $S_{wr} \geq 0.294$ (Apply the reference-Scaled Average Bioequivalence (RSABE) method).

Definitions Introduction of S_{wr}

There is the following conversion relationship between S_{wr}^2 and $CV_w\%$:

$$CV_w\% = \sqrt{e^{S_{wr}^2} - 1} .$$

Therefore, S_{wr}^2 can be used to judge whether the experimental drug belongs to the highly variable drug.

Calculation for S_{wr} can be conducted as follows:

$$S_{wr}^2 = \frac{\sum_{i=1}^m \sum_{j=1}^{n_i} (D_{ij} - \bar{D}_i)^2}{2(n - m)}$$

Where:

i = number of sequences m used in the study

[m=3 for partially replicated design: TRR, RTR, and RRT;

m=2 for fully replicated design: TRTR and RTRT]

j = number of subjects within each sequence

T = Test product

R = Reference product

$$D_{ij} = R_{ij1} - R_{ij2}$$

(where 1 and 2 represent replicate reference treatments)

$$\bar{D}_i = \frac{\sum_{j=1}^{n_i} D_{ij}}{n_i}$$

$$n = \sum_{i=1}^m n_i$$

(i.e. total number of subjects used in the study, while n_i is number of subjects used in sequence i)

Branch Macro to Determine S_{wr} :

Part A: partial reference-replicated 3-way design

1. Define the following quantities:

T_{ij} = the observation on T for subject j within sequence i

R_{ijk} = kth observation ($k = 1$ or 2) on R for subject j within sequence i

$$I_{ij} = T_{ij} - \frac{R_{ij1} + R_{ij2}}{2} \quad \text{and} \quad D_{ij} = R_{ij1} - R_{ij2}$$

I_{ij} is the difference between a subject's (specifically subject j within sequence i) observation on T and the mean of the subject's two observations on R, while D_{ij} is the difference between a subject's two observations on R.

2. Determine I_{ij} and D_{ij}

```
*calculate Iij;
if cmiss(latt, lat1r, lat2r)=0 then ilat=latt-(0.5*(lat1r+lat2r));
*calculate Dij;
if cmiss(lat1r, lat2r)=0 then dlat=lat1r-lat2r;
```

latt: the logarithmic value of the test product

lat1r: the logarithmic value of the first reference product

lat2r: the logarithmic value of the second reference product

3. Intermediate analysis - dlat

```
ods select none;
ods output overallanova=dglm1;
proc glm data=scavbe;
  by parcat1n parcat1 paramn paramcd;
  class trtseqan;
  model dlat=trtseqan;
run;
ods output close;
ods select all;

data dglm1_1;
  set dglm1;
  where source="误差";
  s2wr=ms/2;
  swr=sqrt(s2wr);
  call symputx(cats(paramcd, "swr"), swr);
run;
```

Part B: fully replicated 4-way design

1. Define the following quantities:

T_{ijk} = kth observation ($k = 1$ or 2) on T for subject j within sequence i

R_{ijk} = kth observation ($k = 1$ or 2) on R for subject j within sequence i

$$I_{ij} = \frac{T_{ij1} + T_{ij2}}{2} - \frac{R_{ij1} + R_{ij2}}{2} \quad \text{and} \quad D_{ij} = R_{ij1} - R_{ij2}$$

I_{ij} is the difference between the mean of a subject's (specifically subject j within sequence i) two observations on T and the mean of the subject's two observations on R, while D_{ij} is the difference between a subject's two observations on R.

2. Determine I_{ij} and D_{ij}

```
*calculate Iij;  
if cmiss(lat1t, lat2t, lat1r, lat2r)=0 then ilat=0.5*(lat1t+lat2t-lat1r-lat2r);  
*calculate Dij;  
if cmiss(lat1r, lat2r)=0 then dlat=lat1r-lat2r;
```

lat1t: the logarithmic value of the first test product

lat2t: the logarithmic value of the first test product

lat1r: the logarithmic value of the first reference product

lat2r: the logarithmic value of the second reference product

3. Intermediate analysis – dlat

```
ods select none;  
ods output covparms=dout1 estimates=dout2;  
proc mixed data=scavbe;  
  by parcat1n parcat1 paramn paramcd;  
  class trtseqan;  
  model dlat=trtseqan/dfm=satterth;  
  estimate 'average' intercept 1 trtseqan 0.5 0.5/e cl alpha=0.1;  
run;  
ods output close;  
ods select all;  
  
data dout1_1;  
  set dout1;  
  s2wr=estimate/2;  
  swr=sqrt(s2wr);  
  call symputx(cats(paramcd, "swr"), swr);  
run;
```

Step 2: Find least squares geometric mean, ratio of least squares geometric mean, and 90% CI of GMR when $S_{wr} < 0.294$.

```

ods select none;
ods output lsmeans=lsmeans estimates=estimates covparms=res;
proc mixed data=adpp0(where=(paramcd in (&Swrless294_1.)));
  by parcatln parcatl paramn paramcd;
  class usubjid trtseqan trtan aperiod;
  model aval_lg=trtseqan trtan aperiod / ddfm=satterth;
  random trtan/type=fa0(2) sub=usubjid g;
  repeated/grp=trtan sub=usubjid;
  lsmeans trtan / cl alpha=0.05;
  estimate '受试制剂/参比制剂' trtan 1 -1/cl alpha=0.10;
run;
ods output close;
ods select all;

data part2_lsgem;
  set lsmeans;
  gm=exp(estimate);
run;

data part3_gemo;
  set estimates;
  ratio=exp(estimate)*100;
  low=exp(lower)*100;
  upp=exp(upper)*100;
run;

data part4_intracv;
  set res;
  where covparm="Residual";
  intracv=sqrt(exp(estimate)-1)*100;
run;

```

Step 3: Get 95% upper confidence bound, and find least squares geometric mean, and ratio of least squares geometric mean when $S_{wr} \geq 0.294$ (Apply the reference-Scaled Average Bioequivalence (RSABE) method).

Definitions Introduction of 95% upper confidence bound:

$$(\bar{Y}_T - \bar{Y}_R)^2 - \theta_{S_{WR}}^2$$

Where:

- \bar{Y}_T and \bar{Y}_R are the means of the ln-transformed PK endpoint (AUC and/or C_{max}) obtained from the BE study for the test and reference products, respectively
- $\theta = \left(\frac{\ln(1.25)}{\sigma_{W0}}\right)^2$
- And $\sigma_{W0} = 0.25$ (regulatory limit)

The above formula is converted into a SAS processable calculation procedure:

- When number of sequences equal to three:

```

ods select none;
ods output overallanova=iglm1 estimates=iglm2;
proc glm data=scavbe;
  by parcat1n parcat1 paramn paramcd;
  class trtseqan;
  model ilat=trtseqan/clparm alpha=0.1;
  estimate 'average' intercept 1 trtseqan 0.3333333333 0.3333333333 0.3333333333;
run;
ods output close;
ods select all;

```

- When number of sequences equal to four

```

ods select none;
ods output covparms=iout1 estimates=iout2;
proc mixed data=scavbe;
  by parcat1n parcat1 paramn paramcd;
  class trtseqan;
  model ilat=trtseqan/ddfm=satterth;
  estimate 'average' intercept 1 trtseqan 0.5 0.5/e cl alpha=0.1;
run;
ods output close;
ods select all;

```

- Calculate the result from the output

```

x=estimate**2-stderr**2;
boundx=(max((abs(lowercl)), (abs(uppercl))))**2;
dfd=df;
theta=((log(1.25))/0.25)**2;
y=-theta*s2wr;
boundy=y*dfd/cinv(0.95, dfd);
critbound=(x+y)+sqrt(((boundx-x)**2)+((boundy-y)**2));

```

The %dasforBE macro about Data Analysis for Bioequivalence

The %dasforBE macro includes all of the above steps and supports input parameters, the name of the input data set, the arm number, parameter about whether to contain the category variable, and the name of the output data set. With these parameters, macros can better handle data structures, better realize the function, and can be presented appropriately.

Discussion

For the test product to be bioequivalent to the reference product, both of the following conditions must be satisfied for each PK parameter tested:

- a. the 95% upper confidence bound must be ≤ 0

AND

- b. the point estimate of the Test/Reference geometric mean ratio must fall within [0.80, 1.25].

For drugs with uneven or even steep exposure dose-effect curves, such as ticagrelor, dabigatran, etc., even if the within-subject coefficient of variation of the pharmacokinetic parameter is greater than or equal to 30%, it is not recommended to use the RSABE method to widen the equivalence judgment criteria, to avoid safety risks that may occur in some patients due to increased exposure. Therefore, the statistical method in this paper can only

be used as a statistical reference.

On the macro side, although it supports a cross design was designed for three cycles of partial repetition or four cycles of complete repetition, the input dataset structure and applicable statistical models are limited to a certain extent. What this macro is practical for us is easy to adjust the corresponding part of the macro program for the requirements of different data structures, statistical models, or display modes.

Conclusion

This innovative SAS macro select appropriate statistical methods (RSABE or ABE) to evaluate the equivalence according to the within-subject coefficient of variation, and supports a cross design was designed for three cycles of partial repetition or four cycles of complete repetition, and flexibly preprocess various BDS data or the similar structure of BDS data. In a nutshell, this paper provides a scientific and convenient model for bioequivalence statistics.

Reference

- [1] "Technical Guidelines for Research on Bioequivalence of Highly Variable Drugs." NMPA 2018.
- [2] "Draft Guidance on Progesterone." FDA 2011.
- [3] W. G. Howe (1974), "Approximate Confidence Limits on the Mean of X+Y Where X and Y are Two Tabled Independent Random Variables," *Journal of the American Statistical Association*, 69 (347): 789-794.

Acknowledgement

Thanks for the support from the statistics department of my company, especially thanks to Yumei Peng, Yu Hu, Bin Hu, Xuhua Deng, Haiqiang Tang, Kejing Liu, and Shuangbin Ji for their guidance and advice.

Contact information

Name: Chuying Dai

E-mail: chuying.dai@crmedicon.com

Appendix I: SAS code for macro %dasforBE

➤ Macro Introduction

```
* Program Name      : dasforBE.sas
* Programmer       : Chuying Dai
* Date Created     : 2021-07-30
* Description      : The SAS Macro about Data Analysis for Bioequivalence
* Parameter Setting : in=the name of the input data set,
                   armn=the arm or sequence number,
                   parcat1n=the category variable name "parcat1n" or keep empty (when there is no category),
                   out=the name of the output data set
```

➤ The main macro

```
%macro dasforbe(in=,armn=,parcat1n=,out=);
```

```
*Step1:Data pre-processing and get general parameters (mean and cv) for each treatment groups;
```

```
data adpp0;
```

```

length trt $50;
set &in.;
*add the filter criteria for the dataset;
where besfl="是" and paramcd in ("AUCIFO" "AUCLST" "CMAX");

*determine whether it contains parcat1 variable;
%if "&parcat1n."="" %then %do;
    parcat1n=1;
    parcat1="";
%end;

*tag test or reference groups in order;
%if &armn.=3 %then %do;
    if trta="受试制剂" then do;
        trt="T";
        trtn=1;
    end;
    else if (trtseqa in ("R-T-R" "R-R-T") and aperiod=1) or (trtseqa in ("T-R-R") and
aperiod=2) then do;
        trt="R1";
        trtn=2;
    end;
    else if (trtseqa in ("R-T-R" "T-R-R") and aperiod=3) or (trtseqa in ("R-R-T") and
aperiod=2) then do;
        trt="R2";
        trtn=3;
    end;
%end;
%else %if &armn.=4 %then %do;
    if (trtseqa in ("T-R-T-R") and aperiod=1) or (trtseqa in ("R-T-R-T") and aperiod=2)
then do;
        trt="T1";
        trtn=1;
    end;
    else if (trtseqa in ("T-R-T-R") and aperiod=3) or (trtseqa in ("R-T-R-T") and
aperiod=4) then do;
        trt="T2";
        trtn=2;
    end;
    else if (trtseqa in ("T-R-T-R") and aperiod=2) or (trtseqa in ("R-T-R-T") and
aperiod=1) then do;
        trt="R1";
        trtn=3;
    end;

```

```

        else if (trtseqa in ("T-R-T-R") and aperiod=4) or (trtseqa in ("R-T-R-T") and
aperiod=3) then do;
            trt="R2";
            trtn=4;
        end;
    %end;
    *generate logarithmic analysis value;
    if aval ne . then aval_lg=log(aval);
run;

```

```

proc sort data=adpp0;
    by parcat1n parcat1 paramn paramcd trtn trt;
quit;

```

```

proc univariate data=adpp0 noprint;
    by parcat1n parcat1 paramn paramcd trtn trt;
    var aval;
    output out=part1_meansv mean=mean cv=cv;
quit;

```

*Step2:Determine Swr, the within-subject standard deviation of the reference product, for the pharmacokinetic parameters AUC and Cmax;

```

%if &armn.=3 %then %do;
    %deterswr3();
%end;
%else %if &armn.=4 %then %do;
    %deterswr4();
%end;
%put aucifoswr=&aucifoswr.;
%put auclstswr=&auclstswr.;
%put cmaxswr=&cmaxswr.;
*Gategorize parameters with Swr;
%let Swrless294_I=;
%let Swrgrer294_I=;
%if %sysevalf(&aucifoswr.<0.294) %then %do;
    %let Swrless294_I=&Swrless294_I. 'AUCIFO';
%end;
%else %if %sysevalf(&aucifoswr.>=0.294) %then %do;
    %let Swrgrer294_I=&Swrgrer294_I. 'AUCIFO';
%end;
%if %sysevalf(&auclstswr.<0.294) %then %do;
    %let Swrless294_I=&Swrless294_I. 'AUCLST';
%end;
%else %if %sysevalf(&auclstswr.>=0.294) %then %do;

```

```

    %let Swrgrer294_I=&Swrgrer294_I. 'AUCLST';
%end;
%if %sysevalf(&cmaxswr.<0.294) %then %do;
    %let Swrless294_I=&Swrless294_I. 'CMAX';
%end;
%else %if %sysevalf(&cmaxswr.>=0.294) %then %do;
    %let Swrgrer294_I=&Swrgrer294_I. 'CMAX';
%end;
%put Swrless294_I=&Swrless294_I.;
%put Swrgrer294_I=&Swrgrer294_I.;

```

*Step3:Find least squares geometric mean, and ratio of least squares geometric mean when Swr<0.294;

```

%if "&Swrless294_I." ne "" %then %do;
    ods select none;
    ods output lsmeans=lsmeans estimates=estimates covparms=res;
    proc mixed data=adpp0(where=(paramcd in (&Swrless294_I)));
        by parcat1n parcat1 paramn paramcd;
        class usubjid trtseqan trtan aperiod;
        model aval_lg=trtseqan trtan aperiod / ddfm=satterth;
        random trtan/type=fa0(2) sub=usubjid g;
        repeated/grp=trtan sub=usubjid;
        lsmeans trtan / cl alpha=0.05;
        estimate 'Test/Reference' trtan 1 -1/cl alpha=0.10;
    run;
    ods output close;
    ods select all;

    data part2_lsgem;
        set lsmeans;
        gm=exp(estimate);
        trtn=trtan;
        %if &armn.=4 %then %do;
            if trtan=2 then trtn=3;
        %end;
    run;

    data part3_gemo;
        set estimates;
        ratio=exp(estimate)*100;
        low=exp(lower)*100;
        upp=exp(upper)*100;
        trtn=1;
    run;

```

```

data part4_intracv;
  set res;
  where covparm="Residual";
  intracv=sqrt(exp(estimate)-1)*100;
  trtn=input(compress(group,,"kd"),best.);
  %if &armn.=4 %then %do;
    if input(compress(group,,"kd"),best.)=2 then trtn=3;
  %end;
run;
%end;

*Step4: Get upper limit of 95%CI for the boundary value standard of the RSABE,
and find least squares geometric mean, and ratio of least squares geometric mean
when Swr >= 0.294;
%if "&Swrgrer294_1." ne "" %then %do;

  %if &armn.=3 %then %do;
    *Intermediate analysis - ilat;
    ods select none;
    ods output overallanova=iglm1 estimates=iglm2;
    proc glm data=scavbe;
      by parcat1n parcat1 paramn paramcd;
      class trtseqan;
      model ilat=trtseqan/clparm alpha=0.1;
      estimate 'average' intercept 1 trtseqan 0.3333333333 0.3333333333
0.3333333333;
    run;
    ods output close;
    ods select all;

    data iglm2_1;
      set iglm2;
      x=estimate**2-stderr**2;
      boundx=(max((abs(lowercl)),(abs(uppercl))))**2;
    run;

    *From the above parameters, calculate the final Swr and 95% upper confidence
bound;
    data critbound;
      merge iglm2_1 dglm1_1;
      by parcat1n parcat1 paramn;
      dfd=df;
      theta=((log(1.25))/0.25)**2;

```

```

        y= -theta*s2wr;
        boundy=y*dfd/cinv(0.95,dfd);
        critbound=(x+y)+sqrt(((boundx-x)**2)+((boundy-y)**2));
run;

*Adjust the structure for final display;
data part5_critbound;
    set critbound;
    trtn=1;
    output;
    trtn=2;
    output;
run;

proc sort data=part5_critbound;
    by parcat1n parcat1 paramn trtn;
run;
%end;
%else %if &armn.=4 %then %do;
    *Intermediate analysis - ilat;
    ods select none;
    ods output covparms=iout1 estimates=iout2;
    proc mixed data=scavbe;
        by parcat1n parcat1 paramn paramcd;
        class trtseqan;
        model ilat=trtseqan/ddfm=satterth;
        estimate 'average' intercept 1 trtseqan 0.5 0.5/e cl alpha=0.1;
run;
ods output close;
ods select all;

data iout2_1;
    set iout2;
    x=estimate**2-stderr**2;
    boundx=(max((abs(lower)),(abs(upper))))**2;
run;

data dout2_1;
    set dout2;
    dfd=df;
run;

*From the above parameters, calculate the final Swr and 95% upper confidence
bound;

```

```

data critbound;
  merge iout2_1 dout1_1 dout2_1;
  by parcat1n parcat1 paramn;
  theta=((log(1.25))/0.25)**2;
  y=-theta*s2wr;
  boundy=y*dfd/cinv(0.95,dfd);
  critbound=(x+y)+sqrt(((boundx-x)**2)+((boundy-y)**2));
run;

*Adjust the structure for final display;
data part5_critbound;
  set critbound;
  trtn=1;
  output;
  trtn=3;
  output;
run;

proc sort data=part5_critbound;
  by parcat1n parcat1 paramn trtn;
run;

%end;

ods select none;
ods output lsmeans=lsmeans_h estimates=estimates_h covparms=res_h;
proc mixed data=adpp0(where=(paramcd in (&Swrgrer294_l.))) method=type3;
  by parcat1n parcat1 paramn paramcd;
  class usubjid trtseqan trtan aperiod;
  model aval_lg=trtseqan trtan aperiod / ddfm=kenwardroger;
  random usubjid(trtseqan);
  lsmeans trtan / cl alpha=0.05;
  estimate '受试制剂/参比制剂' trtan 1 -1/cl alpha=0.10;
run;
ods output close;
ods select all;

data part2_lsgem_h;
  set lsmeans_h;
  gm_h=exp(estimate);
  trtn=trtan;
  %if &armn.=4 %then %do;
    if trtan=2 then trtn=3;
  %end;

```

```

run;

data part3_gemo_h;
  %if &armn.=3 %then %do;
    set iglm2;
  %end;
  %else %if &armn.=4 %then %do;
    set iout2;
  %end;
  ratio_h=exp(estimate)*100;
  trtn=1;
run;
%end;

*Step5:Result presentation in output dataset;
data &out.;
  length col1-col12 $200;
  merge part;;
  by parcat1n parcat1 paramn trtn;
  col1=parcat1;
  if paramcd in ("AUCLST") then col2="AUC@{sub 0-t} (h*ng/mL)";
  if paramcd in ("AUCIFO") then col2="AUC@{sub 0-∞} (h*ng/mL)";
  if paramcd in ("CMAX") then col2="C@{sub max} (ng/mL)";
  col3=trtn;
  col4=strip(put(mean,10.2));
  col5=strip(put(cv,10.1));
  %if "&Swrless294_l." ne "" %then %do;
    if swr<0.294 then do;
      col6=strip(put(gm,12.2));
      col7=strip(put(ratio,12.2));
      col8=strip(put(low,12.2));
      col9=strip(put(upp,12.2));
      col10=strip(put(intracv,12.1));
      col12="";
    end;
  %end;
  %if "&Swrgrer294_l." ne "" %then %do;
    if swr>=0.294 then do;
      col6=strip(put(gm_h,12.2));
      col7=strip(put(ratio_h,12.2));
      if trt="T" then col8="NA";
      if trt="T" then col9="NA";
      if trt in ("T" "R1") then col10="NA";
      if trt="R1" then col12=strip(put(critbound,16.4));
    end;
  %end;

```

```

        end;
    %end;
    if trt="R1" then col11=strip(put(swr,12.3));
    %if "&parcat1n."="" %then %do;
        keep col2-col12;
        rename
            col2=col1 col3=col2 col4=col3 col5=col4 col6=col5 col7=col6
            col8=col7 col9=col8 col10=col9 col11=col10 col12=col11
        ;
    %end;
    %else %do;
        keep col;
    %end;
run;

%mend dasforbe;

```

➤ **The branch macro**

```

*****Branch Macro:Determine Swr*****;
%macro deterswr3();
*Determine lij and Dij;
data scavbe;
    merge adpp0(where=(trtn=1) rename=(aval_lg=latt))
          adpp0(where=(trtn=2) rename=(aval_lg=lat1r))
          adpp0(where=(trtn=3) rename=(aval_lg=lat2r));
    by parcat1n parcat1 paramn paramcd usubjid;
    *calculate lij;
    if cmiss(latt,lat1r,lat2r)=0 then ilat=latt-(0.5*(lat1r+lat2r));
    *calculate Dij;
    if cmiss(lat1r,lat2r)=0 then dlat=lat1r-lat2r;
run;

*Intermediate analysis - dlat;
ods select none;
ods output overallanova=dglm1;
proc glm data=scavbe;
    by parcat1n parcat1 paramn paramcd;
    class trtseqan;
    model dlat=trtseqan;
run;
ods output close;
ods select all;

*Calculate the final Swr;

```

```

data dglm1_1;
  set dglm1;
  where source="误差";
  s2wr=ms/2;
  swr=sqrt(s2wr);
  call symputx(cats(paramcd,"swr"),swr);
run;

*Adjust the structure for final display;
data part5_swr;
  set dglm1_1;
  trtn=1;
  output;
  trtn=2;
  output;
run;

proc sort data=part5_swr;
  by parcat1n parcat1 paramn trtn;
run;
%mend deterswr3;

%macro deterswr4();
data scavbe;
  merge adpp0(where=(trtn=1) rename=(aval_lg=lat1t))
        adpp0(where=(trtn=2) rename=(aval_lg=lat2t))
        adpp0(where=(trtn=3) rename=(aval_lg=lat1r))
        adpp0(where=(trtn=4) rename=(aval_lg=lat2r));
  by parcat1n parcat1 paramn paramcd usubjid;
  *calculate lij;
  if cmiss(lat1t,lat2t,lat1r,lat2r)=0 then ilat=0.5*(lat1t+lat2t-lat1r-lat2r);
  *calculate Dij;
  if cmiss(lat1r,lat2r)=0 then dlat=lat1r-lat2r;
run;

*Intermediate analysis - dlat;
ods select none;
ods output covparms=dout1 estimates=dout2;
proc mixed data=scavbe;
  by parcat1n parcat1 paramn paramcd;
  class trtseqan;
  model dlat=trtseqan/ddfm=satterth;
  estimate 'average' intercept 1 trtseqan 0.5 0.5/e cl alpha=0.1;

```

```
run;
ods output close;
ods select all;

*Calculate the final Swr;
data dout1_1;
  set dout1;
  s2wr=estimate/2;
  swr=sqrt(s2wr);
  call symputx(cats(paramcd,"swr"),swr);
run;

*Adjust the structure for final display;
data part5_sw;
  set dout1_1;
  trtn=1;
  output;
  trtn=3;
  output;
run;

proc sort data=part5_sw;
  by parcat1n parcat1 paramn trtn;
run;
%mend deterswr4;
```